

Math 4530
Monday April 15

Weierstrass-Enneper procedure, both representations

```
> restart:with(plots):  
Warning, the name changecoords has been redefined
```


First we do the WE-1 representation which uses general gauss map g . This is a simple adaptation of the WE-2 procedure on page 249 of Oprea. If you had a particular domain you were interested in constructing a minimal surface over you could write a more complicated procedure to compute line integrals, over whatever curves you wanted. This is how you would have to approach cases where the indefinite integration procedure would run into trouble because your domain wasn't rectangular, or worse, not simply connected.

```
> #Procedure which uses general gauss map g, adapted from Oprea page  
249  
WE1:=proc(f,g)  
local Z1,Z2,Z3, X1,X2,X3,X;  
Z1:=int(f*(1-g^2),z);  
Z2:=int(I*f*(1+g^2),z);  
Z3:=int(2*f*g,z);  
X1:=evalc(Re(subs(z=u+I*v,Z1)));  
X2:=evalc(Re(subs(z=u+I*v,Z2)));  
X3:=evalc(Re(subs(z=u+I*v,Z3)));  
X:=simplify([X1,X2,X3]);  
end;
```

Test it on Enneper. We did this example by hand at the end of class on Friday

```
> Enne1:=WE1(-z^2,-1/z);  
plot3d(Enne1,u=-1..1,v=-1..1,color=black,style=wireframe);
```

$$Enne1 := \left[-\frac{1}{3}u^3 + uv^2 + u, u^2v - \frac{1}{3}v^3 + v, u^2 - v^2 \right]$$



You shouldn't be fooled by how easily this one turned out; in many examples Maple will not be as good at simplifying the formula for $X(u,v)$ as you are.

We will talk in class about the second WE-representation, in which you change variables from z to $\tau = g(z)$. In the new variables $G(\tau) = \tau$, so you can think of this as a special case of WE1. If you are changing from WE2 to WE1, the change from (f,g) to (F, id) is a little tricky though, see class notes of Enneper example.

```
> #procedure for second version of Weierstrass formula
WE2:=proc(F)
  WE1(F, z);
end:
```

Test it on the Enneper. From class notes, when $G(z)=z$ then $F(z)=-1/z^4$!

```
> Enne2:=WE2(-1/(z^4));
#what a mess
```

Enne2:=

$$\left[-\frac{1}{3} \frac{(3u^4 - u^2 + 6u^2v^2 + 3v^2 + 3v^4)u}{(u^2 + v^2)(u^4 + 2u^2v^2 + v^4)}, \frac{1}{3} \frac{(3u^4 + 3u^2 + 6u^2v^2 - v^2 + 3v^4)v}{(u^2 + v^2)(u^4 + 2u^2v^2 + v^4)}, \frac{u^2 - v^2}{u^4 + 2u^2v^2 + v^4} \right]$$

```
> simplify(subs({u=-x/(x^2+y^2), v=y/(x^2+y^2)}, Enne2));
#substituting u+iv = g(z+iy)=-1/z, does recover old
parameterization
```

$$\left[-\frac{1}{3}x(x^2 - 3 - 3y^2), \frac{1}{3}y(3x^2 + 3 - y^2), x^2 - y^2 \right]$$


```
> Enne3:=WE2(1);
#the easiest way to get Enneper! (Oprea 7.3.10)
```

$$Enne3 := \left[-\frac{1}{3}u^3 + uv^2 + u, -u^2v + \frac{1}{3}v^3 - v, u^2 - v^2 \right]$$

[Another example, the Catenoid:

```
> Cat1:=WE1(-exp(-z)/2,-exp(z));
plot3d(Cat1,u=-1..1,v=0..2*Pi,color=black,style=wireframe);
```

$$Cat1 := \left[\frac{1}{2} \cos(v) (1 + e^{(2u)}) e^{(-u)}, \frac{1}{2} \sin(v) (1 + e^{(2u)}) e^{(-u)}, u \right]$$



[Whereas:

```
> Cat2:=WE2(1/(2*z^2));
```

$$Cat2 := \left[-\frac{1}{2} \frac{u(u^2 + v^2 + 1)}{u^2 + v^2}, -\frac{1}{2} \frac{v(u^2 + v^2 + 1)}{u^2 + v^2}, \frac{1}{2} \ln(u^2 + v^2) \right]$$

```
> simplify(subs({u=-exp(x)*cos(y),v=-exp(x)*sin(y)},Cat2));
#substituting back in that u+iv = g(x+iy) will recover old formula
```

$$\left[\frac{1}{2} \cos(y) (e^{(2x)} + 1) e^{(-x)}, \frac{1}{2} \sin(y) (e^{(2x)} + 1) e^{(-x)}, \frac{1}{2} \ln(e^{(2x)}) \right]$$